

第4章 補間と関数近似

数値計算では、関数 $u(x)$ は通常 計算点の列 $x_0, x_1, x_2, \dots, x_N$ 上の値 $u_0, u_1, u_2, \dots, u_N$ として離散的 (discrete) に与えられる。このようなデータに対し点列の間の u の値を求めることを補間または内挿 (interpolation) といい、点列の外側の u の値を求めることを外挿 (extrapolation) という (かつては補外ともいわれた)。また離散データを近似する連続関数 $u(x)$ を求めることを関数近似という。関数近似は幾何学的には、与えられた点列を通る滑らかな曲線を引くことである。滑らかな線の標準的なものは、2 点を通る直線、3 点を通る放物線、一般に $n+1$ 個の点を通る n 次多項式である。近似関数 (approximate functions) としては通常 これらの多項式 (polynomials) が用いられるが、数値解法では 解析解と違い 1 区間または数区間ごとに近似関数は異なるものになる。多項式の次数を上げれば、滑らかなデータでは近似の実質的精度も上がるが、ゆらぎを含むデータでは名目的精度は上がっても曲線が波打つことになる。このようなデータに対しても補間の実質的精度を確保するために各種の補間法が考えられている。更に実験データのようにばらつきのあるデータに対しては、最小 2 乗法を用い、与えられたデータを平均的に満足する近似関数が求められる。関数 $u(x)$ が不連続性 (discontinuities) をもつ場合には高次多項式を振動を抑える対策を施して用いるか、別途 不連続性を考慮することが必要である。近似関数 $u(x)$ が決定されれば、補間値だけでなく微分値や積分値も容易に求めることができる。本章には補間に関してのみ述べる。微分値や積分値を数値的に求めることを数値微分または数値積分というが、これらについては章を改めて述べることにする。

4.1 Lagrange 補間多項式

関数 $u(x)$ 上に任意に取られる $n+1$ 個の計算点の列 u_0, u_1, \dots, u_n を通る x の n 次多項式を

$$p_n(x) = a_0 + a_1x + a_2x^2 + \dots + a_nx^n \equiv \sum_{\nu=0}^n a_\nu x^\nu \quad (4.1)$$

とする。ただし a_ν は未定係数で、 $n+1$ 個の計算点の条件

$$u_j = p_n(x_j) = \sum_{\nu=0}^n a_\nu x_j^\nu \quad (j = 0, 1, \dots, n) \quad (4.2a)$$

あるいは

$$\begin{pmatrix} 1 & x_0 & \dots & x_0^n \\ 1 & x_1 & \dots & x_1^n \\ \dots & \dots & \dots & \dots \\ 1 & x_n & \dots & x_n^n \end{pmatrix} \begin{pmatrix} a_0 \\ a_1 \\ \vdots \\ a_n \end{pmatrix} = \begin{pmatrix} u_0 \\ u_1 \\ \vdots \\ u_n \end{pmatrix} \quad (4.2b)$$

から一意的に決定される。それは式 (4.2) が未定係数 a_ν の連立 1 次方程式で、その係数行列が Vandermonde の行列で各式が 1 次独立になっているからである。式 (4.2) から求めた a_ν を式 (4.1) に代入すれば Lagrange 補間多項式が導かれる。以下では これとは異なる方法で Lagrange 補間多項式を導くことにする。

式 (4.1) は $p_n(x)$ と a_ν が 1 次の同次関係にあり, また式 (4.2) は u_j と a_ν が 1 次の同次関係にあることを示している. したがって $p_n(x)$ と u_j も 1 次の同次関係にあり次のように置くことができる.

$$p_n(x) = u_0 b_0(x) + u_1 b_1(x) + u_2 b_2(x) + \cdots + u_n b_n(x) \equiv \sum_{k=0}^n b_k(x) u_k \quad (4.3)$$

ただし $b_k(x)$ は x の高々 n 次の多項式である. この式もまた $n+1$ 個の計算点の条件

$$u_j = \sum_{k=0}^n b_k(x_j) u_k \quad (j = 0, 1, 2, \dots, n)$$

を満足すべきである. この式の両辺の u_j の係数を等置すれば次の関係が得られる.

$$b_k(x_j) = \delta_{kj} = \begin{cases} 0 & (j \neq k) \\ 1 & (j = k) \end{cases}$$

これより n 次多項式 $b_k(x)$ は, n 個の異なる零点 $x_0, \dots, x_{k-1}, x_{k+1}, \dots, x_n$ を持ち, また $b_k(x_k) = 1$ であることが分かる. したがって $b_k(x)$ は次のように決定することができる.

$$b_k(x) = \frac{(x-x_0) \cdots (x-x_{k-1})(x-x_{k+1}) \cdots (x-x_n)}{(x_k-x_0) \cdots (x_k-x_{k-1})(x_k-x_{k+1}) \cdots (x_k-x_n)}$$

Lagrange 補間多項式 (Lagrangean interpolation polynomial) は, これらの係数を式 (4.3) に代入したもので次式で与えられる.

$$p_n(x) = \sum_{k=0}^n \frac{L_k(x)}{L_k(x_k)} u_k \quad (4.4a)$$

ただし

$$L_k(x) \equiv (x-x_0)(x-x_1) \cdots (x-x_{k-1})(x-x_{k+1}) \cdots (x-x_n) \quad (4.4b)$$


ここでよく使われる $n = 1, 2, 3$ の場合, すなわち 1 次式, 2 次式, 3 次式による補間について述べる. $n = 1$ の場合には,

$$u(x) \approx p_1(x) = \frac{x-x_1}{x_0-x_1} u_0 + \frac{x-x_0}{x_1-x_0} u_1$$

あるいは

$$u(\alpha) = (1-\alpha)u_i + \alpha u_{i+1} + O(\Delta x^2)$$

ただし $\alpha = (x-x_i)/\Delta x_{i+1/2}$, $\Delta x_{i+1/2} = x_{i+1}-x_i$, のように表すこともできる. この補間式は x の 1 次式で, $x = x_0$ ならば $u = u_0$, $x = x_1$ ならば $u = u_1$ になり, 正しい補間式であることが容易に確認できる.



$n = 2$ の場合には ,

$$u(x) = \frac{(x-x_1)(x-x_2)}{(x_0-x_1)(x_0-x_2)}u_0 + \frac{(x-x_0)(x-x_2)}{(x_1-x_0)(x_1-x_2)}u_1 + \frac{(x-x_0)(x-x_1)}{(x_2-x_0)(x_2-x_1)}u_2 + O(\Delta x^3)$$

また

$$u(\alpha) = \sum_{j=-1}^1 C_j^2(\alpha) u_{i+j} + O(\Delta x^3) \quad (4.5a)$$

$$C_{-1}^2(\alpha) = \frac{\alpha(\alpha-1)}{m^-(m^-+1)}, \quad C_0^2(\alpha) = -\frac{(\alpha+m^-)(\alpha-1)}{m^-}, \quad C_1^2(\alpha) = \frac{(\alpha+m^-)\alpha}{m^-+1} \quad (4.5b)$$

ただし $\alpha = (x-x_i)/\Delta x_{i+1/2}$, $m^- = \Delta x_{i-1/2}/\Delta x_{i+1/2}$, $\Delta x_{i+1/2} = x_{i+1} - x_i$, のように表すこともできる . これらの式は x の 2 次式で , $\alpha = -m^-, 0, 1$ で $u = u_{i-1}, u_i, u_{i+1}$ になるので正しい補間式である . 等間隔の計算点 ($m^- = 1$) に対しては

$$u(\alpha) = \frac{1}{2}\alpha(\alpha-1)u_{i-1} - (\alpha^2-1)u_i + \frac{1}{2}(\alpha+1)\alpha u_{i+1} + O(\Delta x^3) \quad (4.6)$$

$n = 3$ の場合には ,

$$u(x) = \frac{(x-x_1)(x-x_2)(x-x_3)}{(x_0-x_1)(x_0-x_2)(x_0-x_3)}u_0 + \frac{(x-x_0)(x-x_2)(x-x_3)}{(x_1-x_0)(x_1-x_2)(x_1-x_3)}u_1 \\ + \frac{(x-x_0)(x-x_1)(x-x_3)}{(x_2-x_0)(x_2-x_1)(x_2-x_3)}u_2 + \frac{(x-x_0)(x-x_1)(x-x_2)}{(x_3-x_0)(x_3-x_1)(x_3-x_2)}u_3 + O(\Delta x^4)$$

また

$$u(\alpha) = \sum_{j=-1}^2 C_j^3(\alpha) u_{i+j} + O(\Delta x^4) \quad (4.7a)$$

$$C_{-1}^3(\alpha) = -\frac{\alpha(\alpha-1)(\alpha-1-m^+)}{m^-(m^-+1)(m^-+1+m^+)}, \quad C_0^3(\alpha) = \frac{(\alpha+m^-)(\alpha-1)(\alpha-1-m^+)}{m^-(1+m^+)}, \\ C_1^3(\alpha) = -\frac{(\alpha+m^-)\alpha(\alpha-1-m^+)}{(m^-+1)m^+}, \quad C_2^3(\alpha) = \frac{(\alpha+m^-)\alpha(\alpha-1)}{(m^-+1+m^+)(1+m^+)m^+} \quad (4.7b)$$

ただし $m^\pm = \Delta x_{i+1/2 \pm 1}/\Delta x_{i+1/2}$, $\Delta x_{i+1/2} = x_{i+1} - x_i$ である . この補間公式は通常 $0 < \alpha < 1$ の範囲で用いられるものである . また α すなわち x の 3 次式で , 4 つの計算点の条件 $u(x_{i+j}) = u_{i+j}$ ($j = -1, 0, 1, 2$) を満足しているので一応正しいと言える¹ . 等間隔の計算点 ($m^\pm = 1$) に対しては

$$u(\alpha) = \sum_{k=-1}^2 C_k^3(\alpha) u_{i+k} + O(\Delta x^4) \quad (4.8a)$$

$$C_{-1}^3(\alpha) = -\alpha(\alpha-1)(\alpha-2)/6, \quad C_0^3(\alpha) = (\alpha+1)(\alpha-1)(\alpha-2)/2, \\ C_1^3(\alpha) = -(\alpha+1)\alpha(\alpha-2)/2, \quad C_2^3(\alpha) = (\alpha+1)\alpha(\alpha-1)/6 \quad (4.8b)$$

上記の補間公式の最後の項は , 打ち切り誤差 (truncation error) と呼ばれ , これらの式の精度を表すもので , 実際の計算では無視される . $O(\Delta x^4)$ はこの誤差が Δx の 4 乗の大きさ (order) であることを意味する . 打ち切り誤差 e^t の大きさは Taylor 展開をもとに評価される . 式 (4.7) の両辺に Taylor 展開

$$u(\alpha) = u_i + \alpha h u_i' + \frac{1}{2!} \alpha^2 h^2 u_i'' + \frac{1}{3!} \alpha^3 h^3 u_i''' + \frac{1}{4!} \alpha^4 h^4 u_i^{(4)} + \dots \quad (4.9)$$

¹印刷物に載っている式は必ずしも正確とは言えないので , このような点を確認してから利用するのが無難である .

$$\begin{aligned}
u_{i-1} &= u_i - m^- h u'_i + \frac{1}{2!} (m^-)^2 h^2 u''_i - \frac{1}{3!} (m^-)^3 h^3 u'''_i + \frac{1}{4!} (m^-)^4 h^4 u^{(4)}_i + \dots \\
u_{i+1} &= u_i + h u'_i + \frac{1}{2!} h^2 u''_i + \frac{1}{3!} h^3 u'''_i + \frac{1}{4!} h^4 u^{(4)}_i + \dots \\
u_{i+2} &= u_i + (1+m^+) h u'_i + \frac{1}{2!} (1+m^+)^2 h^2 u''_i + \frac{1}{3!} (1+m^+)^3 h^3 u'''_i + \frac{1}{4!} (1+m^+)^4 h^4 u^{(4)}_i + \dots
\end{aligned} \tag{4.10}$$

ただし $h = \Delta x_i$, を代入し両辺を比較すれば , 打切り誤差 e_t が次のように求められる .

$$e_t = \frac{1}{4!} (\alpha + m^-) \alpha (\alpha - 1) (\alpha - 1 - m^+) h^4 u^{(4)}_i + \dots \tag{4.11}$$

数値計算で用いられる離散近似式は一般に打切り誤差を含むと言われるが , 打切りの意味は離散近似式の精度を Taylor 展開を基に評価するとき , この誤差が Taylor 級数の打切られた項で表されるからである . あるいは離散近似式がある項までで打切った Taylor 級数で作られているからである . 実際の計算で重要なのは , 打切り誤差よりも実質的精度で , これは計算結果から一目瞭然のものである . Taylor 級数が収束する解析関数ないしはそれに近い関数では打切り誤差の小さい高精度の式を用いれば , 実質的精度も上がるが , 不連続性のある関数では Taylor 級数が収束しないので , その収束を前提とする打切り誤差で精度を評価すること自体誤りである . なお数値計算で桁数を打切ったために生ずる誤差は丸めの誤差 (round-off error) と言われ , これは大型の連立 1 次方程式を解く際などに問題になるものである .

ここで少し異なる観点から Lagrange 補間多項式を議論しよう . 今 3 次補間式を 1 次補間式 + 補正項の形で表すことにする . 式 (4.8) の係数を 1 次補間式のものと同補正項のものに分ければ , C_{-1}^3, C_0^3 はそのまま , C_0^3, C_1^3 は次のようになる .

$$C_0^3 = 1 - \alpha + \alpha(\alpha - 1) \frac{\alpha - 1 + m^- - m^+}{m^- (1 + m^+)}, \quad C_1^3 = \alpha - \alpha(\alpha - 1) \frac{\alpha + m^- - m^+}{(m^- + 1)m^+}$$

また式 (4.8) は次のように書くことができる .

$$u(\alpha) = u_i + \alpha \Delta u_i + \alpha(\alpha - 1) \sum_{j=-1}^2 \tilde{C}_j^3 u_{i+j} \tag{4.12}$$

$$\tilde{C}_{-1}^3 = -\frac{\alpha - 1 - m^+}{m^- (m^- + 1)(m^- + 1 + m^+)}, \quad \tilde{C}_0^3 = \frac{\alpha - 1 + m^- - m^+}{m^- (1 + m^+)}, \tag{4.13}$$

$$\tilde{C}_1^3 = -\frac{\alpha + m^- - m^+}{(m^- + 1)m^+}, \quad \tilde{C}_2^3 = \frac{\alpha + m^-}{(m^- + 1 + m^+)(1 + m^+)m^+} \tag{4.14}$$

補正項は $\alpha = 0, 1$ で値 0 であるから , このように因子 $\alpha(\alpha - 1)$ を持つことになる .

更にこの 3 次補間式を 2 次補間式 + 補正項の形にもっていく . 点 $i-1, i, i+1$ を通る 2 次補間式 (4.5) を採ることにすれば , 3 次補間式 (4.8) の係数 C_{-1}^3, C_0^3, C_1^3 は次のようになる .


$$\begin{aligned}
C_{-1}^3 &= \frac{\alpha(\alpha - 1)}{m^- (m^- + 1)} - \frac{\alpha(\alpha - 1)(\alpha + m^-)}{m^- (m^- + 1)(m^- + 1 + m^+)} \\
C_0^3 &= -(\alpha - 1) - \frac{\alpha(\alpha - 1)}{m^-} + \frac{\alpha(\alpha - 1)(\alpha + m^-)}{m^- (1 + m^+)} \\
C_1^3 &= \underbrace{\alpha}_{1 \text{ 次補間}} + \underbrace{\frac{\alpha(\alpha - 1)}{m^- + 1}}_{2 \text{ 次補間}} + \underbrace{\frac{\alpha(\alpha - 1)(\alpha + m^-)}{(m^- + 1)m^+}}_{3 \text{ 次補正項}}
\end{aligned}$$

3次補正項は, $\alpha = -m^-, 0, 1$ で値0であるから, このように因子 $(\alpha+m^-)\alpha(\alpha-1)$ を持つことになる. 結局, 3次補間公式 (4.7) は次のように表すことができる.

$$u(\alpha) = u_i + \alpha \Delta u_{i+1/2} + \frac{1}{2} \alpha(\alpha-1) \left\{ \Delta^2 u_i + \frac{1}{3} (\alpha+m^-) \Delta^3 u_{i+1/2} \right\} \quad (4.15)$$

ただし $\Delta u_{i+1/2}$, $\Delta^2 u_i$, $\Delta^3 u_{i+1/2}$ は不等間隔計算点における1階, 2階, 3階の差分で次のように定義される.

$$\begin{array}{l} u_{i-1} \\ u_i \\ u_{i+1} \\ u_{i+2} \end{array} \left. \begin{array}{l} \Delta u_{i-1/2} = \frac{u_i - u_{i-1}}{m^-} \\ \Delta u_{i+1/2} = u_{i+1} - u_i \\ \Delta u_{i+3/2} = \frac{u_{i+2} - u_{i+1}}{m^+} \end{array} \right\} \left. \begin{array}{l} \Delta^2 u_i = \frac{\Delta u_{i+1/2} - \Delta u_{i-1/2}}{(m^-+1)/2} \\ \Delta^2 u_{i+1} = \frac{\Delta u_{i+3/2} - \Delta u_{i+1/2}}{(1+m^+)/2} \end{array} \right\} \Delta^3 u_{i+1/2} = \frac{\Delta^2 u_{i+1} - \Delta^2 u_i}{(m^-+1+m^+)/3}$$



1階差分に Taylor 展開の式 (4.10) を代入すれば

$$\begin{aligned} \Delta u_{i-1/2} &= hu'_i - \frac{1}{2!} m^- h^2 u''_i + \frac{1}{3!} (m^-)^2 h^3 u'''_i - \frac{1}{4!} (m^-)^3 h^4 u^{(4)}_i + \dots \\ \Delta u_{i+1/2} &= hu'_i + \frac{1}{2!} h^2 u''_i + \frac{1}{3!} h^3 u'''_i + \frac{1}{4!} h^4 u^{(4)}_i + \dots \\ \Delta u_{i+3/2} &= hu'_i + \frac{1}{2!} (2+m^+) h^2 u''_i + \frac{1}{3!} \{3+3m^+(m^+)\} h^3 u'''_i \\ &\quad + \frac{1}{4!} \{4+6m^++4(m^+)^2+(m^+)^3\} h^4 u^{(4)}_i + \dots \end{aligned}$$

更にこれらの式から2階差分, 3階差分の Taylor 展開式は次のようになる.

$$\begin{aligned} \Delta^2 u_i &= 2 \left\{ \frac{1}{2!} h^2 u''_i + \frac{1}{3!} (1-m^-) h^3 u'''_i + \frac{1}{4!} \{1-m^-(m^-)^2\} h^4 u^{(4)}_i + \dots \right\} \\ \Delta^2 u_{i+1} &= 2 \left\{ \frac{1}{2!} h^2 u''_i + \frac{1}{3!} (2+m^+) h^3 u'''_i + \frac{1}{4!} \{3+3m^+(m^+)\} h^4 u^{(4)}_i + \dots \right\} \\ \Delta^3 u_{i+1/2} &= 6 \left\{ \frac{1}{3!} h^3 u'''_i + \frac{1}{4!} (2-m^-+m^+) h^4 u^{(4)}_i + \dots \right\} \end{aligned}$$

式 (4.15) の両辺に式 (4.9) と上記の差分式の Taylor 展開式を代入すれば, h^3 までの項は相殺され,

$$\frac{1}{4!} \alpha^4 h^4 u^{(4)}_i + \dots = \frac{1}{4!} [\alpha + \alpha(\alpha-1) \{1-m^-(m^-)^2 + (\alpha+m^-)(2-m^-+m^+)\}] h^4 u^{(4)}_i + \dots + e^t$$

先に示した3次 Lagrange 補間多項式の打ち切り誤差, 式 (4.11) はこのようにして求めることができる. 一方次のように簡単に求めることもできる. 打ち切り誤差は, 計算点 $(\alpha = -m^-, 0, 1, 1+m^+)$ で0になるから, 因子 $(\alpha+m^-)\alpha(\alpha-1)(\alpha-1-m^+)$ を持ち α の4次式になる. 式 (4.7) の左辺に Taylor 展開式 (4.9) を代入し, また打ち切り誤差をこの因子に係数を掛けたもので表せば,

$$u_i + \alpha h u'_i + \dots + \frac{1}{4!} \alpha^4 h^4 u^{(4)}_i = [3 \text{ 次 Lagrange 補間多項式}] + a(\alpha+m^-)\alpha(\alpha-1)(\alpha-1-m^+)$$

係数 a は α^4 の項を比較することによって求められ, $a = h^4 u^{(4)}_i / 4!$ となる.

4.2 Newton の補間公式

等間隔計算点の場合に使用できるいくつかの差分形補間公式について述べる．Newton 前進補間公式は計算領域の左端近傍の補間に，また Newton 後退補間公式は右端近傍の補間に良く登場する．しかしながらこれらの補間公式は，特に端近傍を精度良く補間するものではなく，Lagrange 補間公式と見かけは違うが等価なものである．差分形補間公式には，ほかに Bessel 補間公式，Gauss 補間公式，Stirling 補間公式などがある．これらの補間公式をあえて使う必要はないが，文献等に出てくることもあるのでここには簡単に結果のみ示すことにする．

4.2.1 Newton 前進補間公式

Lagrange 補間多項式はもちろん端の区間の補間にも外挿にも使うことができる．一方 等間隔計算点の場合には，左端の区間 $x_0 \leq x \leq x_1$ の補間に Newton 前進補間公式 (Newton's forward interpolation formula) が広く用いられている．この補間公式には前進差分 (forward-differences) が使われるので，まずその定義から述べることにする．

$$\begin{aligned} 1 \text{ 階差分} \quad u_{i+1} - u_i &= \Delta u_i \\ 2 \text{ 階差分} \quad \Delta u_{i+1} - \Delta u_i &= \Delta^2 u_i \\ &\dots\dots\dots \\ k \text{ 階差分} \quad \Delta^{k-1} u_{i+1} - \Delta^{k-1} u_i &= \Delta^k u_i \end{aligned}$$

これらの差分の関係は差分表にまとめられ，次のようになる．

x	u	Δ	Δ^2	Δ^3
x_0	u_0			
x_1	u_1	Δu_0		
x_2	u_2	Δu_1	$\Delta^2 u_0$	
x_3	u_3	Δu_2	$\Delta^2 u_1$	$\Delta^3 u_0$
		Δu_3	$\Delta^2 u_2$	$\Delta^3 u_1$

逆に前進差分を関数 u_i で表せば次のようになる．

$$\begin{aligned} \Delta u_i &= -u_i + u_{i+1}, \\ \Delta^2 u_i &= u_i - 2u_{i+1} + u_{i+2}, \\ \Delta^3 u_i &= -u_i + 3u_{i+1} - 3u_{i+2} + u_{i+3}, \quad \dots\dots\dots \\ \Delta^k u_i &= \sum_{\nu=0}^k (-1)^{k-\nu} \binom{k}{\nu} u_{i+\nu}, \quad \text{ただし } \binom{k}{n} = \frac{k!}{n!(k-n)!} \text{ は 2 項係数} \end{aligned} \tag{4.16}$$

Newton の前進補間公式では，まず関数 $u(x)$ を次の n 次多項式で近似する．

$$\begin{aligned} p_n(x) &= a_0 + a_1(x-x_0) \\ &\quad + a_2(x-x_0)(x-x_1) + \dots \\ &\quad + a_n(x-x_0)(x-x_1)\dots(x-x_{n-1}) \\ &= a_0 + \sum_{k=1}^n a_k(x-x_0)(x-x_1)\dots(x-x_{k-1}) \end{aligned} \tag{4.17}$$

ただし a_0, a_1, \dots, a_n は未定係数で, 次の計算点の条件を満足するように決定される.

$$u_j = p_k(x_j) = a_0 + \sum_{k=1}^j a_k(x_j - x_0)(x_j - x_1) \cdots (x_j - x_{k-1}) \quad (j = 0, 1, \dots, n)$$

あるいは

$$\begin{pmatrix} 1 & & & & 0 \\ 1 & h & & & \\ 1 & 2h & 2h^2 & & \\ \vdots & \vdots & \vdots & \ddots & \\ 1 & nh & n(n-1)h^2 \cdots & n!h^n & \end{pmatrix} \begin{pmatrix} a_0 \\ a_1 \\ a_2 \\ \vdots \\ a_n \end{pmatrix} = \begin{pmatrix} u_0 \\ u_1 \\ u_2 \\ \vdots \\ u_n \end{pmatrix}$$

この式は a_k の連立 1 次方程式で, 下三角行列であるから係数 a_k は上から順に決定することができる.

$$\begin{aligned} a_0 &= u_0 \\ a_1 &= \frac{1}{h}(u_1 - u_0) = \frac{1}{h} \Delta u_0 \\ a_2 &= \frac{1}{2!h^2}(u_2 - 2u_1 + u_0) = \frac{1}{2!h^2} \Delta^2 u_0 \\ &\dots\dots\dots \\ a_n &= \dots\dots\dots = \frac{1}{n!h^n} \Delta^n u_0 \end{aligned} \quad (4.18)$$

Newton 前進補間公式は x の多項式 (4.17) とその係数の式 (4.18) で与えられる. この補間公式はまた, 左端の点を x_i とすれば, $x = x_i + \alpha h$ で定義される α を用い次のように表すこともできる.

$$\begin{aligned} u(\alpha) &= u_i + \alpha \Delta u_i + \frac{1}{2!} \alpha(\alpha-1) \Delta^2 u_i + \cdots + \frac{1}{n!} \alpha(\alpha-1) \cdots (\alpha-n+1) \Delta^n u_i + O(h^{n+1}) \\ &= \sum_{k=0}^n \binom{\alpha}{k} \Delta^k u_i + O(h^{n+1}) \end{aligned} \quad (4.19)$$

この補間公式は等間隔計算点 $x_i, x_{i+1}, x_{i+2}, \dots$ の関数値が既知のときに, 主に区間 $x_i < x < x_{i+1}$ ($0 < \alpha < 1$) の関数値を補間する目的で用いられる. また区間 $x < x_i$ ($\alpha < 0$) に適用し外挿に用いられることもある. 補間が妥当に行われているか否かは, 高次差分の値のばらつきと $u_i, \Delta u_i, \Delta^2 u_i, \dots$ の 0 への収束性から判定することができる.

問 Newton 前進形補間公式が等間隔計算点の Lagrange 補間多項式と等価なものであることを示せ.

等間隔の Lagrange 補間多項式 (4.4) は, $x = x_0 + \alpha h$ で定義される α を用いれば次のようになる.

$$\begin{aligned} &\sum_{\nu=0}^n \frac{(x-x_0) \cdots (x-x_{\nu-1})(x-x_{\nu+1}) \cdots (x-x_n)}{(x_\nu-x_0) \cdots (x_\nu-x_{\nu-1})(x_\nu-x_{\nu+1}) \cdots (x_\nu-x_n)} u_\nu \\ &= \sum_{\nu=0}^n \frac{(-1)^{n-\nu}}{\nu!(n-\nu)!} \alpha(\alpha-1) \cdots (\alpha-\nu+1)(\alpha-\nu-1) \cdots (\alpha-n) u_\nu \end{aligned}$$


一方 Newton 前進補間公式 (4.19) は

$$\begin{aligned} \sum_{k=0}^n \binom{\alpha}{k} \Delta^k u_0 &= \sum_{k=0}^n \frac{1}{k!} \alpha(\alpha-1) \cdots (\alpha-k+1) \sum_{\nu=0}^k (-1)^{k-\nu} \binom{k}{\nu} u_\nu \\ &= \sum_{\nu=0}^n \frac{1}{\nu!} \alpha(\alpha-1) \cdots (\alpha-\nu+1) \left[\sum_{k=\nu}^n \frac{(-1)^{k-\nu}}{(k-\nu)!} (\alpha-\nu) \cdots (\alpha-k+1) \right] u_\nu \end{aligned}$$

この式の k に関する総和はもともとの意味を考えれば次のようになり，前の項から順に足算すれば容易に 1 つの項に纏めることができる．

$$\begin{aligned} & \sum_{k=\nu}^n \frac{(-1)^{k-\nu}}{(k-\nu)!} (\alpha-\nu) \cdots (\alpha-k+1) \\ &= 1 - (\alpha-\nu) + \frac{1}{2!} (\alpha-\nu)(\alpha-\nu-1) - \cdots + \frac{(-1)^{n-\nu}}{(n-\nu)!} (\alpha-\nu) \cdots (\alpha-n+1) \\ &= \frac{(-1)^{n-\nu}}{(n-\nu)!} (\alpha-\nu-1) \cdots (\alpha-n) \end{aligned}$$

これより Newton 前進形補間公式と等間隔計算点の Lagrange 補間多項式は等価なものであることが分かる．
なお簡単に $n+1$ 個のデータ u_j を通る n 次多項式は一意的に決まるからということもできよう．



4.2.2 Newton 後退補間公式

前項の Newton 前進補間公式は計算領域の左端近傍の補間に用いられるもので，右端近傍の補間には Newton 後退補間公式が用いられる．後退差分 (backward-differences) は

$$\begin{aligned} 1 \text{ 階差分} & \quad u_i - u_{i-1} &= \nabla u_i \\ 2 \text{ 階差分} & \quad \nabla u_i - \nabla u_{i-1} &= \nabla^2 u_i \\ & \quad \dots\dots\dots \\ k \text{ 階差分} & \quad \nabla^{k-1} u_i - \nabla^{k-1} u_{i-1} &= \nabla^k u_i \end{aligned}$$

のように定義される．また後退差分の関係は次の差分表のようになる．

x	u	∇	∇^2	∇^3
x_7	u_7	} ∇u_8	} $\nabla^2 u_8$	} $\nabla^3 u_9$
x_8	u_8			
x_9	u_9	} ∇u_{10}	} $\nabla^2 u_{10}$	} $\nabla^3 u_{10}$
x_{10}	u_{10}			

逆に後退差分を関数 u_i で表せば次のようになる .

$$\begin{aligned}\nabla u_i &= u_i - u_{i-1}, \\ \nabla^2 u_i &= u_i - 2u_{i-1} + u_{i-2}, \\ \nabla^3 u_i &= u_i - 3u_{i-1} + 3u_{i-2} - u_{i-3}, \quad \dots\dots \\ \nabla^k u_i &= \sum_{\nu=0}^k (-1)^\nu \binom{k}{\nu} u_{i-\nu}\end{aligned}\tag{4.20}$$

Newton の後退補間公式 (Newton's backward interpolation formula) は , 関数 $u(x)$ の n 次多項式近似 $p_n(x)$ とその係数 $a_0, a_1, a_2, \dots, a_n$ で与えられる .

$$p_n(x) = a_0 + \sum_{k=1}^n a_k (x-x_i)(x-x_{i-1}) \cdots (x-x_{i-k+1})\tag{4.21a}$$

$$a_0 = u_i, \quad a_k = \frac{1}{k!h^k} \nabla^k u_i \quad (k = 1, 2, \dots, n)\tag{4.21b}$$

ただし x_i は右端の点で , この式は通常 $x_{i-1} < x < x_i$ の補間に用いられる . この補間公式は , $x = x_i + \alpha h$ と置けば次のようになる .

$$\begin{aligned}u(x) &= u_i + \alpha \nabla u_i + \frac{1}{2!} \alpha(\alpha+1) \nabla^2 u_i + \cdots + \frac{1}{n!} \alpha(\alpha+1) \cdots (\alpha+n-1) \nabla^n u_i + O(h^{n+1}) \\ &= \sum_{k=0}^n \binom{\alpha+k-1}{k} \nabla^k u_i + O(h^{n+1})\end{aligned}\tag{4.22}$$

この式は通常 $-1 < \alpha < 0$ の補間に用いられる .

4.2.3 Gauss 前進補間公式 , Gauss 後退補間公式 , Stirling 補間公式 , Bseesl 補間公式

ここではこれらの補間公式を中心差分 (central-differences) を用いて示すことにする . 中心差分の演算子は δ または Δ で表され , 次のように定義される .

$$\begin{aligned}\Delta u_{i+1/2} &= u_{i+1} - u_i &= \Delta u_i &= \nabla u_{i+1}, \\ \Delta^2 u_i &= \Delta u_{i+1/2} - \Delta u_{i-1/2} &= \Delta^2 u_{i-1} &= \nabla^2 u_{i+1}, \\ \Delta^3 u_{i+1/2} &= \Delta^2 u_{i+1} - \Delta^2 u_i &= \Delta^3 u_{i-1} &= \nabla^3 u_{i+2}, \\ \Delta^{2m} u_i &= \Delta^{2m-1} u_{i+1/2} - \Delta^{2m-1} u_{i-1/2} &= \Delta^{2m} u_{i-m} &= \nabla^{2m} u_{i+m}, \\ \Delta^{2m+1} u_{i+1/2} &= \Delta^{2m} u_{i+1} - \Delta^{2m} u_i &= \Delta^{2m+1} u_{i-m} &= \nabla^{2m+1} u_{i+m+1}\end{aligned}\tag{4.23}$$

上式にはまた中心差分と前進差分 , 後退差分の関係も示してある . 中心差分の差分表を次に示す .

x	u	Δ	Δ^2	Δ^3
x_{i-1}	u_{i-1}	}	}	}
x_i	u_i			
x_{i+1}	u_{i+1}	}	}	}
x_{i+2}	u_{i+2}			

Gauss 前進補間公式は，

$$u(x) = a_0 + a_1(x-x_0) + a_2(x-x_0)(x-x_1) + a_3(x-x_0)(x-x_1)(x-x_2) + \dots$$

と置き計算点の条件 $u_0 = u(x_0)$, $u_1 = u(x_1)$, $u_{-1} = u(x_{-1})$, $u_2 = u(x_2)$, $u_{-2} = u(x_{-2})$, ... を順に満足させることによって導かれ次のようになる．

$$u(x) \doteq u_i + \binom{\alpha}{1} \Delta u_{i+1/2} + \binom{\alpha}{2} \Delta^2 u_i + \binom{\alpha+1}{3} \Delta^3 u_{i+1/2} + \binom{\alpha+1}{4} \Delta^4 u_i + \dots \tag{4.24}$$

ただし $x = x_i + \alpha h$ である．同様にして次の Gauss 後退補間公式が得られる．

$$u(x) \doteq u_i + \binom{\alpha}{1} \Delta u_{i-1/2} + \binom{\alpha+1}{2} \Delta^2 u_i + \binom{\alpha+1}{3} \Delta^3 u_{i-1/2} + \binom{\alpha+2}{4} \Delta^4 u_i + \dots \tag{4.25}$$

Gauss 前進補間公式 (4.24) (図 4.1 の細線) と Gauss 後退補間公式 (4.25) (図 4.1 の二重線) を平均すれば，次の Stirling 補間公式が得られる．

$$u(x) \doteq u_i + \alpha \frac{\Delta u_{i+1/2} + \Delta u_{i-1/2}}{2} + \frac{1}{2!} \alpha^2 \Delta^2 u_i + \frac{1}{3!} \alpha(\alpha^2 - 1) \frac{\Delta^3 u_{i+1/2} + \Delta^3 u_{i-1/2}}{2} + \frac{1}{4!} \alpha^2(\alpha^2 - 1) \Delta^4 u_i + \dots$$

$$(-1/2 \leq \alpha \leq 1/2) \tag{4.26}$$

また Gauss 前進補間公式 (4.24) と計算点を 1 つシフトした Gauss 後退補間公式 (図 4.1 の太線)

$$u(x) \doteq u_{i+1} + \binom{\alpha-1}{1} \Delta u_{i+1/2} + \binom{\alpha}{2} \Delta^2 u_{i+1} + \binom{\alpha}{3} \Delta^3 u_{i+1/2} + \binom{\alpha+1}{4} \Delta^4 u_{i+1} + \dots$$

を平均すれば，次の Bessel 補間公式が得られる．

$$u(x) \doteq \frac{u_i + u_{i+1}}{2} + (\alpha - \frac{1}{2}) \Delta u_{i+1/2} + \frac{1}{2!} \alpha(\alpha-1) \frac{\Delta^2 u_i + \Delta^2 u_{i+1}}{2} + \frac{1}{3!} \alpha(\alpha-1)(\alpha - \frac{1}{2}) \Delta^3 u_{i+1/2} + \frac{1}{4!} (\alpha^2 - 1) \alpha(\alpha-2) \frac{\Delta^4 u_i + \Delta^4 u_{i+1}}{2} + \dots$$

$$(0 < \alpha < 1) \tag{4.27}$$




図 4.1: 各種の補間公式の関係

4.3 スプライン法

与えられたデータ $u_i = u(x_i)$, $i = 0, 1, 2, \dots, n$ を通る滑らかな曲線を引く問題を考える. Lagrange 補間多項式, 例えば u_0, u_1, u_2, u_3 を通る 3 次式, u_3, u_4, u_5, u_6 を通る 3 次式, を用いれば, 全体の曲線は一般に 3 次式の接点 x_3, x_6, \dots で折れ曲ることになる. また u_0, u_1, u_2, u_3 を通る 3 次式の区間 $[x_1, x_2]$, u_1, u_2, u_3, u_4 を通る 3 次式の区間 $[x_2, x_3]$, を繋げば, 折れ曲りは一般に小さくなるが完全にはなくはない. 他方 全区間を 1 つの高次多項式または Fourier 型級数で近似する古典的方法によれば, 一応解析的に滑らかな曲線は得られるが, 与えられたデータによっては低次多項式に比べ大きく波打つ虞がある. スプライン法は, 2 階微分まで連続な区分的 3 次曲線, 一般には $n+1$ 階微分まで連続な区分的 $2n+1$ 次曲線で近似するものである.

4.3.1 3 次スプライン

普通用いられているものは 3 次スプライン曲線 (cubic spline) で, その基礎式は次の区分的 3 次多項式と接点の条件からなる.

$$S(x) = a_0(1-\alpha)^3 + a_1\alpha^3 + a_2(1-\alpha) + a_3\alpha \quad (0 \leq \alpha \leq 1) \quad (4.28)$$

$$S(x_i) = u_i, \quad S(x_{i+1}) = u_{i+1}; \quad S''(x_i) = u''_i, \quad S''(x_{i+1}) = u''_{i+1}$$

ただし $\alpha = (x-x_i)/\Delta x_i$, $\Delta x_i = x_{i+1}-x_i$ である. 上の 3 次式の 4 つの未定係数を 4 つの接点条件から決定すれば, 次の 3 次スプラインの式を導くことができる.

$$S(x) = \frac{\Delta x_i^2}{6} \left\{ (1-\alpha)^3 u''_i + \alpha^3 u''_{i+1} \right\} + (1-\alpha) \left(u_i - \frac{\Delta x_i^2}{6} u''_i \right) + \alpha \left(u_{i+1} - \frac{\Delta x_i^2}{6} u''_{i+1} \right) \quad (4.29)$$

なお 3 次多項式 $S(x) = \sum_{k=0}^3 a_k \alpha^k$ から始めても同じ結果が得られる. この 3 次スプラインは, 接点条件として変数 u とその 2 階微分 u'' を与えているので, 隣接 3 次曲線との接点 x_i, x_{i+1} で曲線そのものとその曲率が連続になる. 式 (4.29) を x で微分し $\alpha = 0$ または 1 と置けば接点における勾配の式が得られる.

$$u'_i = \frac{u_{i+1}-u_i}{\Delta x_i} - \frac{\Delta x_i}{6} (2u''_i + u''_{i+1}) \quad (4.30a)$$

$$u'_{i+1} = \frac{u_{i+1}-u_i}{\Delta x_i} + \frac{\Delta x_i}{6} (u''_i + 2u''_{i+1}) \quad (4.30b)$$

接点 x_i で両側のスプラインの勾配も連続になるように, 式 (4.30a) と式 (4.30b) の添字 i を $i-1$ に付替えた式を等値すれば, 変数とその 2 階微分の関係を示す次のスプライン公式が導かれる.

$$\frac{\Delta x_{i-1}}{6} u''_{i-1} + \frac{\Delta x_{i-1} + \Delta x_i}{3} u''_i + \frac{\Delta x_i}{6} u''_{i+1} = \frac{u_{i+1}-u_i}{\Delta x_i} - \frac{u_i-u_{i-1}}{\Delta x_{i-1}} \quad (4.31)$$

この式は, $u''_{i\pm 1}$, $u_{i\pm 1}$ の Taylor 展開式を代入すれば, 各辺が

$$\frac{1}{2} (\Delta x_{i-1} + \Delta x_i) \left\{ u''_i + \frac{1}{3} (\Delta x_i - \Delta x_{i-1}) u'''_i \right\} + O(\Delta x^3)$$

のようになり, 3 次精度であることが分かる. また両辺を $(\Delta x_{i-1} + \Delta x_i)/2$ で割算すれば, 各辺は u'' の 2 階差分, スプライン曲線の曲率を表すものになる.

与えられた $I+1$ 個のデータ u_i ($i = 0, 1, \dots, I$) を通る 3 次スプライン曲線を描く手順は次のようになる. (i) 両端の勾配 u'_0, u'_I の値を何らかの方法で与える. (ii) 点 x_1, x_2, \dots, x_{I-1} における式 (4.31) と点

x_0 における式 (4.30a), 点 x_I の式 (4.30b) から曲率 u''_i ($i = 0, 1, \dots, I$) の値を求める. (iii) 3 次スプラインの式 (4.29) で補間スプライン曲線を描く. この 3 次スプラインは変数とその 2 階までの微分すなわち勾配と曲率も連続な曲線である. 本稿では高次スプラインについては省略し, 次に実用的に有用な指数スプラインとばらつきのあるデータにも適用できる最小 2 乗スプラインについて述べることにする.

4.3.2 指数スプライン

式 (4.29) は常微分方程式 $S^{(4)} = 0$ の解で, はり (梁) の曲げ問題のアナロジーでは はりの形状を表している. 上記の 3 次スプライン曲線は両端固定支持, 接点で自由支持の場合に該当する. この曲線が波打ち場合には, はりに適切な張力をかければ波打ちをある程度抑えることができ自然なものになる. はりに張力をかけた場合の常微分方程式は $\tilde{S}^{(4)} - p^2\tilde{S}'' = 0$ となり, その解は指数関数ないしは双曲線関数を含むものになる. それゆえこの場合の曲線は指数スプライン (exponential spline) と呼ばれる.

指数スプラインの基礎式は次のように与えることができる.

$$\begin{aligned} \tilde{S}(x) &= a_0 \sinh((1-\alpha)\Delta x p) + a_1 \sinh(\alpha\Delta x p) + a_2(1-\alpha) + a_3\alpha \quad (0 \leq \alpha \leq 1) \\ \tilde{S}(x_i) &= u_i, \quad \tilde{S}(x_{i+1}) = u_{i+1}; \quad \tilde{S}''(x_i) = u''_i, \quad \tilde{S}''(x_{i+1}) = u''_{i+1} \end{aligned} \quad (4.32)$$

ただし p_i^2 は区間 Δx_i のはりにかかる張力で, この架空のはりには各区間ごとに異なる値を与えることができる. なお $\sinh((1-\alpha)\Delta x p) = \sinh(\Delta x p) \cosh(\alpha\Delta x p) - \cosh(\Delta x p) \sinh(\alpha\Delta x p)$ であるから, 上式は $\cosh(\alpha\Delta x p)$ と $\sinh(\alpha\Delta x p)$ を 1 次結合したものと同等のものである. また接点条件は 3 次スプラインの場合と同様のものである. 上式の未定係数を 4 つの接点条件から決定すれば, 次の指数スプラインの式が導かれる.

$$\begin{aligned} \tilde{S}(x) &= \frac{1}{p_i^2 \sinh \Delta x_i p_i} \{ u''_i \sinh((1-\alpha)\Delta x_i p_i) + u''_{i+1} \sinh(\alpha\Delta x_i p_i) \} \\ &\quad + (1-\alpha) \left(u_i - \frac{u''_i}{p_i^2} \right) + \alpha \left(u_{i+1} - \frac{u''_{i+1}}{p_i^2} \right) \end{aligned} \quad (4.33)$$

3 次スプラインの式 (4.30) と (4.31) に相当の式は次のようになる.

$$u'_0 = \frac{u_1 - u_0}{\Delta x_0} - (d_0 u''_0 + e_0 u''_1), \quad u'_I = \frac{u_I - u_{I-1}}{\Delta x_{I-1}} + (e_{I-1} u''_{I-1} + d_{I-1} u''_I) \quad (4.34)$$

$$e_{i-1} u''_{i-1} + (d_{i-1} + d_i) u''_i + e_i u''_{i+1} = \frac{u_{i+1} - u_i}{\Delta x_i} - \frac{u_i - u_{i-1}}{\Delta x_{i-1}} \quad (i = 1, 2, \dots, I-1) \quad (4.35)$$

ただし

$$d_i = -\frac{1}{\Delta x_i p_i^2} + \frac{1}{p_i} \coth \Delta x_i p_i, \quad e_i = \frac{1}{\Delta x_i p_i^2} - \frac{1}{p_i} \operatorname{cosech} \Delta x_i p_i \quad (4.36)$$

なお x_i, u_i, u'_i, u''_i は接点に定義される量, 他方 $\Delta x_i, p_i, d_i, e_i$ は区間に定義される量である. 指数スプライン曲線も上と同様の手順で描くことができる. 張力 $p_i = 0$ ならば 3 次スプライン, p_i を十分大きく取れば折れ線になる.

問 張力が十分小さいときに指数スプラインが 3 次スプラインに近づくことを示せ.

式 (4.32) の係数を \tilde{a}_0, \dots で示すことにする. 式 (4.28) と (4.29) から $a_0 = \Delta x_i^2 u''_i / 6$, $a_2 = u_i - \Delta x_i^2 u''_i / 6$, また x 小のときに $\sinh \approx x + x^3/6$ であるから, p_i 小のときに式 (4.32) と (4.33) から $\tilde{a}_0 = u''_i / (\phi \sinh P) = u''_i / (\phi P)$, $\tilde{a}_0 P + \tilde{a}_2 = u''_i P / (\phi \sinh P) + u_i - u''_i / \phi = u''_i (1 - P^2/6) / \phi + u_i - u''_i / \phi = u_i - \Delta x_i^2 u''_i / 6$ なる関係が

得られる。ただし $P = \Delta x_i p_i$, $\phi = p_i^2$ である。全く同様の関係が a_1, a_3 と \tilde{a}_1, \tilde{a}_3 の間にも成立する。これより次の関係が導かれる。

$$a_0 = \tilde{a}_0 P^3/6, \quad a_1 = \tilde{a}_1 P^3/6, \quad a_2 = \tilde{a}_0 P + \tilde{a}_2, \quad a_3 = \tilde{a}_1 P + \tilde{a}_3$$

p_i 小のときの式 (4.32), すなわち $\tilde{S}(x) \approx \tilde{a}_0 P^3(1-\alpha)^3/6 + \tilde{a}_1 P^3 \alpha^3/6 + (\tilde{a}_0 P + \tilde{a}_2)(1-\alpha) + (\tilde{a}_1 P + \tilde{a}_3)\alpha$ に上の関係を入れれば式 (4.28) になり, p_i 小のときにこれらの式が等価であることが分かる。

4.3.3 スプライン法のプログラム

前記のスプラインと指数スプラインによる補間式の適用性を次の3つの場合について示す。

$$\begin{aligned} \text{CASE 1:} & \quad x^2 + \frac{1}{2}(1-x), & \quad (-0.6 \leq x \leq 1) \\ \text{CASE 2:} & \quad \frac{1}{x+1}, & \quad (-0.6 \leq x \leq 1) \\ \text{CASE 3:} & \quad \begin{cases} 1 & (-0.6 \leq x \leq 0) \\ \exp(-2x) & (0 < x \leq 1) \end{cases} \end{aligned}$$

CASE 1 は 2 次式, CASE 2 は多項式でない代数式, CASE 3 は勾配の不連続を含む場合である。ここでは, 補間式の接点のデータ u_i を, これらの代数式で表される曲線上に比較的粗い等間隔の刻み $\Delta x = 0.2$ で取り, 接点の中間点における補間値を計算し, 基の曲線に対する偏差 δu を評価することにする。以下に示すプログラムはメインプログラム, スプラインのサブルーチン SPL, 指数スプラインのサブルーチン ESPL などからなる。

```

program MAIN
DIMENSION xi(-3:5),ui(-3:5),x(-6:10),u(-6:10),u0(-6:10),du(-6:10),nj(-3:5)
DATA is,if,ns,nf/-3,5,-6,10/ nj/-6,-4,-2,0,2,4,6,8,10/
OPEN(20,FILE='OUTPUT.dat')
FORALL(i=is:if)xi(i)=.2*i
FORALL(n=ns:nf)x(n)=.1*n
nn=0; 100 nn=nn+1
! Compute data and gradients at both ends of original curve
      GOTO(21,22,23),nn

! Case 1: x^2+(1-x)/2
21 FORALL(i=is:if)ui(i)=xi(i)*xi(i)+(1.-xi(i))/2.
   gs=-1.7; gf=1.5;          GOTO 25
! Case 2: 1/(x+1)
22 FORALL(i=is:if)ui(i)=1./(xi(i)+1.)
   gs=-1./16; gf=-1./4.;    GOTO 25
! Case 3: exp(-2x)
23 FORALL(i=is:0)ui(i)=1.; FORALL(i=1:if)ui(i)=EXP(-2.*xi(i))
   gs=0.; gf=-2.*EXP(-2.*1.)
25 CALL SPL(xi,ui,is,if,gs,gf,x,u,ns,nf,nj)           !select subroutine SPL or ESPL
!25 CALL ESPL(xi,ui,is,if,gs,gf,x,u,ns,nf,nj)
! Compute deviations of spline from original curve
      GOTO(31,32,33),nn

! Case 1: x^2+(1-x)/2
31 FORALL(n=ns:nf)u0(n)=x(n)*x(n)+(1.-x(n))/2.; GOTO 35
! Case 2: 1/(x+1)
32 FORALL(n=ns:nf)u0(n)=1./(x(n)+1.);          GOTO 35
! Case 3: exp(-2x)

```

```

33 FORALL (n=ns:0)u0(n)=1.; FORALL (n=1:nf)u0(n)=EXP(-2.*x(n))
35 FORALL (n=ns:nf)du(n)=u(n)-u0(n)
WRITE(20,'(17F10.5)')(u0(n),n=-6,10)
WRITE(20,'(17F10.5)')( u(n),n=-6,10)
WRITE(20,'(17F10.5)')(du(n),n=-6,10)
IF(nn<3)                                GOTO 100
CLOSE(20)
STOP
END

!***** Interpolation by cubic spline
SUBROUTINE SPL(xi,ui,is,if,gs,gf,x,u,ns,nf,nj)
DIMENSION xi(is:if),ui(is:if),x(ns:nf),u(ns:nf),nj(is:if),dx(-3:5),a(-3:5,3),b(-3:5)
FORALL (i=is:if-1)dx(i)=xi(i+1)-xi(i)                !Delta x
FORALL (i=is:if,j=1:3)a(i,j)=0.
FORALL (i=is:if)b(i)=0.
! Compute u''
cycle_1: DO i=is,if
  IF(i==is)                                GOTO 11
  a(i,1)=dx(i-1)/6.; a(i,2)=dx(i-1)/3.                !coefs of lhs
  b(i)=- (ui(i)-ui(i-1))/dx(i-1)                    !rhs
  IF(i==if)CYCLE cycle_1
  11 a(i,2)=a(i,2)+dx(i)/3.; a(i,3)=a(i,3)+dx(i)/6.    !coefs of lhs
  b(i)=b(i)+(ui(i+1)-ui(i))/dx(i)                    !rhs
ENDDO cycle_1
b(is)=b(is)-gs; b(if)=b(if)+gf                        !add grads
CALL GAUSS3(a,b,is,if)                                !compute u''
! Compute of spline
i=is
DO n=ns,nf-1; IF(n==nj(i+1))i=i+1
  dx2=dx(i)*dx(i)
  al=(x(n)-xi(i))/dx(i); al1=1.-al                    !alpha, 1-alpha
  u(n) = dx2/6.*(al1*al1*al1*b(i)+al*al*al*b(i+1)) &    !compute spline
  +(ui(i)-dx2/6.*b(i))*al1+(ui(i+1)-dx2/6.*b(i+1))*al
ENDDO
u(nf) = ui(if)
END

!***** Interpolation by exponential spline
SUBROUTINE ESPL(xi,ui,is,if,gs,gf,x,u,ns,nf,nj)
DIMENSION xi(is:if),ui(is:if),x(ns:nf),u(ns:nf),nj(is:if),dx(-3:5),a(-3:5,3),b(-3:5), &
  c(-3:5),p(-3:5),d(-3:5),e(-3:5)
FORALL (i=is:if-1)dx(i)=xi(i+1)-xi(i)                !Delta x
FORALL (i=is:if-1)p(i)=.1
n=0; 100 n=n+1
DO i=is,if-1
  xp=dx(i)*p(i); p1=1./p(i)
  d(i)=-p1/xp+p1/TANH(xp); e(i)= p1/xp-p1/SINH(xp)    !compute d & e
ENDDO
FORALL (i=is:if,j=1:3)a(i,j)=0.
FORALL (i=is:if)b(i)=0.
! Compute u'' by eqs(4.25)(4.26)
cycle_1: DO i=is,if
  IF(i==is)                                GOTO 11
  a(i,1)=e(i-1); a(i,2)=d(i-1)                    !coefs of lhs
  b(i)=- (ui(i)-ui(i-1))/dx(i-1)                    !rhs
  IF(i==if)CYCLE cycle_1

```

```

11 a(i,2)=a(i,2)+d(i); a(i,3)=a(i,3)+e(i)           !coefs of lhs
b(i)=b(i)+(ui(i+1)-ui(i))/dx(i)                   !rhs
c(i)=b(i)
ENDDO cycle_1
b(is)=b(is)-gs; b(if)=b(if)+gf                   !add grads
CALL GAUSS3(a,b,is,if)                            !compt u''
! If necessary, increase tension
ip=0
cycle_2: DO i=is+1,if-1
  b0=b(i); c0=2.*c(i)/(dx(i-1)+dx(i))             !spline & numer  curvs
  IF(b0*c0>0..AND..8*ABS(b0)<ABS(c0))CYCLE cycle_2
  p(i-1)=p(i-1)+.1*n*n; p(i)=p(i)+.1*n*n         !increase tension
  ip=ip+1
ENDDO cycle_2
WRITE(20,'(2I10,6F10.1)')n,ip,p
IF(n<10.AND.ip>0)                                GOTO 100
! Compute of spline
i=is
DO n=ns,nf-1; IF(n==nj(i+1))i=i+1
  al=(x(n)-xi(i))/dx(i); al1=1.-al                !alpha, 1-alpha
  xp=dx(i)*p(i); p2=1./p(i)/p(i)
  u(n)=(ui(i)-p2*b(i))*al1+(ui(i+1)-p2*b(i+1))*al & !compt spline
  +p2/SINH(xp)*(b(i)*SINH(al1*xp)+b(i+1)*SINH(al*xp))
ENDDO
u(nf) = ui(if)
END

!***** Subroutine GAUSS3
! Solution of linear eqns with tri-diag matrix by Gaussian elimination
SUBROUTINE GAUSS3(a,b,is,if)
REAL a(is:if,3),b(is:if)
cycle_1: DO i=is,if
  b(i)=b(i)/a(i,2); IF(i==if)CYCLE cycle_1; a(i,3)=a(i,3)/a(i,2)
  b(i+1)=b(i+1)-b(i)*a(i+1,1); a(i+1,2)=a(i+1,2)-a(i,3)*a(i+1,1)
ENDDO cycle_1
DO i=if-1,is,-1
  b(i)=b(i)-b(i+1)*a(i,3)
ENDDO
END

```

上記のプログラムについて簡単に説明する。 $x_i = x_i$, $u_i = u_i$ 、 x , u はここでは等間隔刻み $\Delta x = 0.1$ の x_n または u_n , u_0 は基の曲線上の u の値。また $du = \delta u$ は偏差値で、 n 偶数の接点では 0, n 奇数の中間点では下表に示すようになる。 n_j は接点上の n を格納する配列で、任意点の補間は n_j と $x(n)$ を調整することにより可能である。メインプログラムでは、始めに接点の u_i とその始点 is と終点 if の勾配 g_s, g_f を入力し、サブルーチン SPL または ESPL を引用して補間の計算を実行し、最後に δu などを出力している。

次にサブルーチン SPL について説明する。一般には、始めに与えられた u_i から g_s, g_f を計算しなければならないが、すでにメインプログラムで解析的に求めたものを与えているのでこの部分は省略されている。まず式 (4.30a)(4.31)(4.30b) からなる u_i'' の連立 1 次方程式を Gauss 消去法で解いて接点の曲率 u_i'' を計算する。 a, b はこの連立 1 次方程式の 3 重対角行列の係数または右辺の配列、GAUSS3 はこの連立 1 次方程式を解くサブルーチンである。最後に補間点の $a_1 = \alpha$ を普遍性のある方法で求め、式 (4.29) から補間値 $u(n) = u_n$ を計算している。

指数スプラインのサブルーチン ESPL も同様に作られている。張力 $p = p_i$ は、初め小さい正数を与えるが、計算で求めたスプラインの曲率 $b_0 = u_i''$ が与えられた式 (4.35) の右辺の数値曲率 $c_0 = (u_{i+1} - u_i) / \Delta x_i -$

$(u_i - u_{i-1})/\Delta x_{i-1}$ とかなり違うところでは、違いを減らすために局所的にその値を徐々に増やさなければならぬ。ここでは初めに $p_i = 0.1$ を与えているが、このとき式 (4.33) の指数スプラインは 3 次スプラインに極めて近いものになる。またスプラインの曲率と数値曲率の符号が異なる接点 ($b_0 \times c_0 < 0$)、また同じでもスプラインの曲率の大きさが数値曲率に比べかなり大きい接点 ($0.8|b_0| \geq |c_0$) に対しては、その接点の両側の p_i の値を加速度的に増やしている。と同時に計算も始めからやり直ししている。

計算の結果は下表に示すようになる。exact の欄の数字は基の曲線の中間点における値、SPL と ESPL の欄はスプラインまたは指数スプラインで補間した中間点の偏差値である。CASE 1 の 2 次多項式の場合には、当然のこととはいえ偏差値は非常に小さい。また CASE 2 の多項式で近似し難い代数式の場合には、前方の u の値が加速度的に大きくなる場所で若干の偏差値が生じる。CASE 3 の勾配に不連続のある場合には、不連続の近辺で偏差値は大きくなるが、この場合には指数スプラインによる補間がただのスプラインよりも優れていることが分かる。最下段の ESPL の欄は上記プログラムのスプラインの曲率にかかる係数を 0.4、最大反復数を 16 に変更したもので、多少の改善がみられる。

$x =$		-0.5	-0.3	-0.1	0.1	0.3	0.5	0.7	0.9
CASE 1	exact	1.0000	.7400	.5600	.4600	.4400	.5000	.6400	.8600
	SPL	0	0	0	0	0	0	0	0
	ESPL	0	0	0	0	0	0	0	0
CASE 2	exact	2.0000	1.4286	1.1111	.9091	.7692	.6667	.5882	.5263
	SPL	-46	6	-4	0	0	0	0	0
	ESPL	-46	6	-4	0	-1	0	0	0
CASE 3	exact	1.0000	1.0000	1.0000	.8187	.5488	.3679	.2466	.1653
	SPL	15	-74	280	353	-95	25	-7	1
	ESPL	0	-2	42	196	77	57	40	12
	ESPL	0	0	12	174	99	-50	13	-3




図 4.2: CASE 3 の基の曲線と補間値

4.4 最小2乗法に基づく補間

実験データのようにばらつきのあるデータ u_k を近似する式は、通常 最小2乗法によって求められる。最小2乗法 (least square method) では、どのような式で近似するのが鍵で、 $1/x$ で良く近似できるデータを2次式で近似しようとするのはそのひとつの悪い例である。初めに一つの多項式で近似する場合を取り上げる。最小2乗法では通常、 n 次多項式 (4.1) の $n+1$ 個の未定係数 a_ν は、それよりもかなり多い K 個の条件

$$u_k = \sum_{\nu=0}^n a_\nu x_k^\nu \quad (k = 1, 2, \dots, K) \quad (4.37)$$

から決定される。もちろんこれらの条件は、すべて完全に満足させることはできないので、式 (4.37) の残差の2乗の和

$$S(a_0, a_1, \dots, a_n) = \sum_{k=1}^K \left(\sum_{\nu=0}^n a_\nu x_k^\nu - u_k \right)^2 \quad (4.38)$$

が最小になるように平均的に満足させることになる。式 (4.38) の S は a_μ ($\mu = 0, 1, \dots, n$) の2次式で、 S を極値にする a_μ は S を a_μ で微分した式を0と置いた a_ν の連立1次方程式

$$\sum_{\nu=0}^n c_{\mu\nu} a_\nu = f_\mu \quad (\mu = 0, 1, \dots, n) \quad (4.39a)$$

ただし

$$c_{\mu\nu} = \sum_{k=1}^K x_k^\mu x_k^\nu = \sum_{k=1}^K x_k^{\mu+\nu}, \quad f_\mu = \sum_{k=1}^K x_k^\mu u_k \quad (4.39b)$$

から求められる。この極値が S の最小値であることは、 S の式を a_μ で2階微分すれば $c_{\mu\mu} = \sum_{k=1}^K x_k^\mu x_k^\mu > 0$ となることから明らかである。このようにして求めた a_ν を n 次多項式 (4.1) に用いれば、最小2乗法に基づく補間式が得られる。

3次式 $n = 3$ の場合には、式 (4.39) は次のようになる。

$$\begin{aligned} C_0 a_0 + C_1 a_1 + C_2 a_2 + C_3 a_3 &= f_0 \\ C_1 a_0 + C_2 a_1 + C_3 a_2 + C_4 a_3 &= f_1 \\ C_2 a_0 + C_3 a_1 + C_4 a_2 + C_5 a_3 &= f_2 \\ C_3 a_0 + C_4 a_1 + C_5 a_2 + C_6 a_3 &= f_3 \end{aligned}$$

ただし $C_m = x_1^m + \dots + x_K^m$, $f_\mu = x_1^\mu u_1 + \dots + x_K^\mu u_K$ である。

4.4.1 最小2乗スプライン

与えられたデータが1つの多項式で近似することが適当でない場合には、いくつかの区間に区切り 各区間ごとに別々の多項式で近似することにすれば、一応どのようなデータにも対処できることになる。しかしながらこのようにして得られた近似関数は区分的連続で、一般に区間の接点で不連続になる。不連続を小さくするために、区間をオーバーラップさせて取る方法、各区間でその外側のデータも適宜考慮する方法な

どが考案されている．以下に述べる最小 2 乗スプラインは，計算量は多くなるが，区間の接点で関数自身とその 2 階微分までが連続になり，その決定版とも言えるものである．

スプラインの式は前記の式 (4.29) を用いることにする．

$$S(x) = a_0(1-\alpha)^3 + a_1\alpha^3 + a_2(1-\alpha) + a_3\alpha \quad (4.40)$$

ただし $x = x_i + \alpha\Delta x_i$ ， $\Delta x_i = x_{i+1} - x_i$ である．式 (4.40) を微分すれば次のようになる．

$$S'(x) = \Delta x_i^{-1} \{-3a_0(1-\alpha)^2 + 3a_1\alpha^2 - a_2 + a_3\} \quad (4.41a)$$

$$S''(x) = 6\Delta x_i^{-2} \{a_0(1-\alpha) + a_1\alpha\} \quad (4.41b)$$

未定係数 a_0, a_1 は，接点条件 $S''(x_i) = u''_i$ と $S''(x_{i+1}) = u''_{i+1}$ から直接決定することができる．

$$a_0 = \frac{1}{6}\Delta x_i^2 u''_i, \quad a_1 = \frac{1}{6}\Delta x_i^2 u''_{i+1} \quad (4.42)$$

一方 残りの未定係数 a_2, a_3 は， $S(x)$ が 区間 $[x_i, x_{i+1}]$ のデータ u_1, u_2, \dots, u_K を平均的に満足するように，最小 2 乗法によって決定される．残差の 2 乗の和

$$\sum_{k=1}^K [a_0(1-\alpha_k)^3 + a_1\alpha_k^3 + a_2(1-\alpha_k) + a_3\alpha_k - u_k]^2$$

を a_2 または a_3 で微分しゼロと置けば次式が導かれる．

$$\begin{aligned} C_1 a_2 + C_2 a_3 &= \tilde{u}_i - C_4 a_0 - C_6 a_1, \\ C_2 a_2 + C_3 a_3 &= \tilde{u}_{i+1} - C_5 a_0 - C_7 a_1 \end{aligned} \quad (4.43a)$$

ただし

$$\begin{aligned} C_1 &= \sum_{k=1}^K (1-\alpha_k)^2, & C_2 &= \sum_{k=1}^K (1-\alpha_k)\alpha_k, & C_3 &= \sum_{k=1}^K \alpha_k^2, \\ C_4 &= \sum_{k=1}^K (1-\alpha_k)^4, & C_5 &= \sum_{k=1}^K (1-\alpha_k)^3\alpha_k, & C_6 &= \sum_{k=1}^K (1-\alpha_k)\alpha_k^3, \\ C_7 &= \sum_{k=1}^K \alpha_k^4, & \tilde{u}_i &= \sum_{k=1}^K (1-\alpha_k)u_k, & \tilde{u}_{i+1} &= \sum_{k=1}^K \alpha_k u_k \end{aligned} \quad (4.43b)$$

a_2, a_3 はこの式から次のように求められる．

$$\begin{aligned} a_2 &= -\frac{1}{\Delta} (C_2 \tilde{u}_{i+1} - C_3 \tilde{u}_i + D_2 a_0 + D_4 a_1), \\ a_3 &= \frac{1}{\Delta} (C_1 \tilde{u}_{i+1} - C_2 \tilde{u}_i + D_1 a_0 + D_3 a_1) \end{aligned} \quad (4.44a)$$

ただし

$$\begin{aligned} \Delta &= C_1 C_3 - C_2^2 = \sum_{k=1}^K (1-\alpha_k) \sum_{l=1}^K \alpha_l (\alpha_l - \alpha_k) = \sum_{k=1}^K \sum_{l=1}^K \frac{(\alpha_l - \alpha_k)^2}{2}, \\ D_1 &= C_4 C_2 - C_5 C_1 = \sum_{k=1}^K (1-\alpha_k)^3 \sum_{l=1}^K (1-\alpha_l) (\alpha_l - \alpha_k), \end{aligned}$$

$$D_2 = C_4 C_3 - C_5 C_2 = \sum_{k=1}^K (1-\alpha_k)^3 \sum_{l=1}^K \alpha_l (\alpha_l - \alpha_k) = \sum_{k=1}^K \sum_{l=1}^K \frac{(\alpha_l - \alpha_k)^2}{2} (1 - 3\alpha_k \alpha_l) - D_4, \quad (4.44b)$$

$$D_3 = C_6 C_2 - C_7 C_1 = \sum_{k=1}^K \alpha_k^3 \sum_{l=1}^K (1-\alpha_l)(\alpha_l - \alpha_k),$$

$$D_4 = C_6 C_3 - C_7 C_2 = \sum_{k=1}^K \alpha_k^3 \sum_{l=1}^K \alpha_l (\alpha_l - \alpha_k) = - \sum_{k=1}^K \sum_{l=1}^K \frac{(\alpha_l - \alpha_k)^2}{2} \alpha_k \alpha_l (\alpha_k + \alpha_l)$$

なおこの式の導出過程で次の関係が用いられた .

$$\sum_{k=1}^K \sum_{l=1}^K (\alpha_l - \alpha_k) = 0$$

式 (4.41a) に式 (4.42)(4.44) を用いれば ,

$$u'_i = \frac{r_i \tilde{u}_{i+1} - s_i \tilde{u}_i}{\Delta x_i} - \left(\frac{1}{2} - d_i\right) u''_i - e_i u''_{i+1} \quad (4.45a)$$

$$u'_{i+1} = \frac{r_i \tilde{u}_{i+1} - s_i \tilde{u}_i}{\Delta x_i} + d_i u''_i + \left(\frac{1}{2} - e_i\right) u''_{i+1} \quad (4.45b)$$

ただし

$$r_i = \frac{C_1 + C_2}{\Delta} = \frac{1}{\Delta} \sum_{k=1}^K (1-\alpha_k) = \frac{K}{\Delta} - s_i, \quad s_i = \frac{C_2 + C_3}{\Delta} = \frac{1}{\Delta} \sum_{k=1}^K \alpha_k,$$

$$\begin{aligned} d_i &= \frac{\Delta x_i}{6} \frac{D_1 + D_2}{\Delta} = \frac{\Delta x_i}{6} \frac{1}{\Delta} \sum_{k=1}^K (1-\alpha_k)^3 \sum_{l=1}^K (\alpha_l - \alpha_k) \\ &= e_i + \frac{\Delta x_i}{6} \frac{1}{\Delta} \sum_{k=1}^K \sum_{l=1}^K \frac{(\alpha_l - \alpha_k)^2}{2} 3(-\alpha_k - \alpha_l), \end{aligned} \quad (4.45c)$$

$$\begin{aligned} e_i &= -\frac{\Delta x_i}{6} \frac{D_3 + D_4}{\Delta} = -\frac{\Delta x_i}{6} \frac{1}{\Delta} \sum_{k=1}^K \alpha_k^3 \sum_{l=1}^K (\alpha_l - \alpha_k) \\ &= \frac{\Delta x_i}{6} \frac{1}{\Delta} \sum_{k=1}^K \sum_{l=1}^K \frac{(\alpha_l - \alpha_k)^2}{2} (\alpha_k^2 + \alpha_k \alpha_l + \alpha_l^2) \end{aligned}$$

また式 (4.45a) の u'_i と式 (4.45b) の $i \rightarrow i-1$ と置いた式の u'_i を等値すれば ,

$$d_{i-1} u''_{i-1} + (1 - e_{i-1} - d_i) u''_i + e_i u''_{i+1} = \frac{r_i \tilde{u}_{i+1} - s_i \tilde{u}_i}{\Delta x_i} - \frac{r_{i-1} \tilde{u}_i - s_{i-1} \tilde{u}_{i-1}}{\Delta x_{i-1}} \quad (4.46)$$

なる関係が得られる . 最小 2 乗スプラインは , 式 (4.40) , (4.42) , (4.44) , (4.45) , (4.46) を用い , 3 次スプラインの場合と同様の手順で描くことができる . なお 区間 $[x_i, x_{i+1}]$ の未定係数 a_2, a_3 を決める際には , 通常この区間内のデータ u_1, u_2, \dots, u_K が用られるが , この区間の外側のデータも考慮した方がより良い曲線が得られるともいわれている . 最小 2 乗スプライン曲線は 高さ , 勾配 , 曲率まで連続な区分的 3 次曲線である .

ここで最小 2 乗スプライン法の一つのプログラムを示すことにする . このプログラムは曲線

$$u(x) = 0.5 + 0.5 \exp(-2x) \quad (0 \leq x \leq 1)$$

のまわりに適当に分散するデータを与え , このデータを近似するスプライン曲線を上記の最小 2 乗法によって求めるものである .

```

program MAIN
DIMENSION x(50),u(50),kn(0:5),dx(0:5),un(0:20)
DATA kf,if,nf/50,5,20/
DATA x/.031,.040,.058,.070,.071,.089,.106,.120,.141,.160,.179,.194,.228,.229,.232,.275, &
      .279,.318,.323,.368,.372,.389,.430,.439,.465,.508,.525,.538,.570,.578,.612,.626,.656, &
      .677,.682,.705,.717,.759,.786,.819,.822,.834,.853,.861,.908,.909,.928,.945,.972,.991/
DATA u/.978,.959,.943,.927,.931,.923,.898,.896,.864,.859,.850,.831,.804,.813,.830,.778, &
      .800,.757,.753,.728,.737,.729,.710,.699,.700,.688,.682,.661,.669,.661,.654,.649,.647, &
      .620,.636,.614,.626,.621,.594,.600,.601,.591,.598,.578,.583,.578,.580,.586,.563,.563/
DATA g0,gf/-1.,-.13533/                                !grads at both ends
DATA kn/12,10,8,9,11,0/ dx/5*.2,0./                    !numbers of data in dx
OPEN(20,FILE='OUTPUT.dat')
CALL LSSPL(x,u,kf,g0,gf,kn,dx,if,un,nf)                !compute ls-spline
WRITE(20,'(50F10.5)')x
WRITE(20,'(50F10.5)')u
WRITE(20,'(21F10.5)')un                                !printout ls-spline
CLOSE(20)
CALL LSSPLG(x,u,kf,un,nf)
STOP; END

!***** Interpolation by least square spline
SUBROUTINE LSSPL(x,u,kf,g0,gf,kn,dx,if,un,nf)
DIMENSION x(kf),u(kf),kn(0:if),dx(0:if),un(0:nf),a0(0:5),a1(0:5),a2(0:5),a3(0:5), &
          tu(0:5),ttu(0:5),del(0:5),r(0:5),s(0:5),d(0:5),e(0:5),a(0:5,3),b(0:5)
! Prepare d_i,e_i,r_i,s_i... to compute u''
kb=1; ke=kn(0); xi=0.
DO i=0,if-1; dx0=dx(i)
  DO k=kb,ke; ak=(x(k)-xi)/dx0
    tu(i)=tu(i)+(1.-ak)*u(k)                                !tilde u
    ttu(i+1)=ttu(i+1)+ak*u(k)                              !tildetilde u
    s(i)=s(i)+ak
  DO l=kb,ke; al=(x(l)-xi)/dx0
    ak1=.5*(al-ak)*(al-ak)
    del(i)=del(i)+ak1                                       !Delta
    d(i)=d(i)+ak1*3.*(1.-ak-al)
    e(i)=e(i)+ak1*(ak*ak+ak*al+al*al)
  ENDDO
ENDDO
s(i)=s(i)/del(i)                                           !s_i
r(i)=FLOAT(kn(i))/del(i)-s(i)                               !r_i
d(i)=(d(i)+e(i))*dx0/6./del(i)                             !d_i
e(i)=-e(i)*dx0/6./del(i)                                  !e_i
kb=ke+1; ke=ke+kn(i+1)
xi=xi+dx0
ENDDO
! Determine u''
cycle_1: DO i=0,if
  IF(i==0) GOTO 11
  a(i,1)=d(i-1); a(i,2)=-.5-e(i-1)                          !coefs
  b(i)=- (r(i-1)*ttu(i)-s(i-1)*tu(i-1))/dx0                !rhs
  IF(i==if)CYCLE cycle_1
  11 a(i,2)=a(i,2)+.5-d(i); a(i,3)=e(i)                     !coefs
  b(i)=b(i)+(r(i)*ttu(i+1)-s(i)*tu(i))/dx0                 !rhs
ENDDO cycle_1
b(0)=b(0)-g0; b(if)=b(if)+gf                                !grads at both ends
CALL GAUSS3(a,b,if)                                         !compute u''
! Compute a_0,a_1,a_2,a_3

```

```

kb=1; ke=kn(0); xi=0.
DO i=0,if-1; dx0=dx(i)
  a0(i)=b(i)*dx0*dx0/6. !a_0
  a1(i)=b(i+1)*dx0*dx0/6. !a_1
  C1=0.; C2=0.; C3=0.; D2=0.; D4=0.
  DO k=kb,ke; ak=(x(k)-xi)/dx0
    C1=C1+(1.-ak)*(1.-ak) !C_1
    C2=C2+(1.-ak)*ak !C_2
    C3=C3+ak*ak !C_3
    DO l=kb,ke; al=(x(l)-xi)/dx0
      ak1=.5*(al-ak)*(al-ak)
      D2=D2+ak1*(1.-3.*ak*al) !D_2
      D4=D4-ak1*ak*al*(ak+al) !D_4
    ENDDO
  ENDDO
  D1= 6.*del(i)*d(i)/dx0-D2 !D_1
  D3=-6.*del(i)*e(i)/dx0-D4 !D_3
  a2(i)=- (C2*ttu(i+1)-C3*ttu(i)+D2*a0(i)+D4*a1(i))/del(i) !a_2
  a3(i)= (C1*ttu(i+1)-C2*ttu(i)+D1*a0(i)+D3*a1(i))/del(i) !a_3
  kb=ke+1; ke=ke+kn(i+1)
  xi=xi+dx0
ENDDO
! Compute least square spline and deviations
i=0; xi=0.
DO n=0,nf-1
  al=(.05*FLOAT(n)-xi)/dx(i); al1=1.-al !alpha 1-alpha
  un(n)=a0(i)*al1*al1*al1+a1(i)*al*al*al+a2(i)*al1+a3(i)*al !compute ls-spline
  IF (.05*FLOAT(n+1)>=xi+dx(i)) THEN
    xi=xi+dx(i); i=i+1
  ENDF
ENDDO
un(nf)=a1(if-1)+a3(if-1)
kb=1; ke=kn(0); xi=0.; deva=0.; devr=0.
DO i=0,if-1
  DO k=kb,ke
    al=(x(k)-xi)/dx(i); al1=1.-al !alpha 1-alpha
    du=u(k)-(a0(i)*al1*al1*al1+a1(i)*al*al*al+a2(i)*al1+a3(i)*al) !residuals
    deva=deva+ABS(du)/FLOAT(kf) !absolute mean deviation
    devr=devr+du*du/FLOAT(kf)
  ENDDO
  kb=ke+1; ke=ke+kn(i+1)
  xi=xi+dx(i)
ENDDO
devr=SQRT(devr) !root mean square deviation
WRITE(20,'(2F10.5)')deva,devr
END

```

メインプログラムでは、50組のデータ $x = x_k$, $u = u_k$, 両端の勾配 $g_0 = u'_0 = -1$, $g_f = u'_f = -\exp(-2)$ を与え、また計算領域 $0 \leq x \leq 1$ を 5 区間に等分割し、各区間のデータ数を kn に、区間の長さを dx に入れ、スプライン曲線 $un = u_i$ をサブルーチン LSSPL を引用して計算し、スプライン曲線と 2 つの平均偏差値

$$deva = \frac{1}{K} \sum_{k=1}^K |\delta u|, \quad devr = \left(\frac{1}{K} \sum_{k=1}^K \delta u^2 \right)^{1/2}, \quad \delta u = u_k - S(x_k)$$

を出力している。サブルーチン LSSPL では、接点の曲率 u''_i , スプラインの係数 a_0, a_1, a_2, a_3 , スプライン曲線 u_n の順に計算している。まず u''_i の計算の準備として、式 (4.43b) から $\tilde{u}_i, \tilde{u}_{i+1}$, 式 (4.44b) から Δ ,

式 (4.45c) から s_i, r_i, d_i, e_i を計算 . 式 (4.45a) , (4.46) , (4.45b) からなる連立 1 次方程式を GAUSS3 を用いて解き u'_i を決定 . 次に式 (4.42) から係数 a_0, a_1 を計算 . また式 (4.43b) から C_1, C_2, C_3 , 式 (4.44b) から D_2, D_4 , 式 (4.45c) から D_1, D_3 を求め , 式 (4.44a) から係数 a_2, a_3 を計算 . 最後に式 (4.40) によってスプライン u_n , それから上記の 2 つの平均偏差値を計算している .

計算の結果は下図に示すようになる . 赤点線は基の曲線でその後実線で描いたスプライン曲線によってかなり消されている . なお両端の勾配はここでは上式の微分値を与えたが , 本来は端近傍のデータを眺めて与えるべきものである . 下表は u'_0 の値の不確かさがスプライン曲線にどのような影響を与えるかを調べたものである . 表より u'_0 を無視することは適当でないが , 不確かさの影響は小さく局所的であることが分かる .

u'_0	$x = 0$	0.05	0.1	0.2	0.4
-1.0	.9958	.9531	.9112	.8329	.7179
-0.9	.9947	.9527	.9110	.8329	.7179
0	.9856	.9487	.9086	.8331	.7179




図 4.3: 与えられたデータと最小 2 乗スプライン曲線

4.4.2 最小 2 乗指数曲線

与えられたばらつきのあるデータ u_k を 1 次式と指数関数を重ね合わせた次式で近似する .

$$u(x) = a^x b + c(1-x) + d x \quad (0 \leq x \leq 1) \quad (4.47)$$

ただし a, b, c, d は未定係数で , 与えられたデータを平均的に満足するように最小 2 乗法によって決定される . 残差 $u(x_k) - u_x = \Delta u_k$ の 2 乗の和

$$\sum_{k=1}^K \Delta u_k^2 = \sum_{k=1}^K \{a^{x_k} b + c(1-x_k) + d x_k - u_k\}^2 \quad (4.48)$$

を a, b, c または d で微分し 0 と置けば次式が得られる .

$$\begin{aligned}
 \sum_{k=1}^K f_k^1(a, b, c, d) &= \sum_{k=1}^K \Delta u_k a^{x_k} x_k = 0, \\
 \sum_{k=1}^K f_k^2(a, b, c, d) &= \sum_{k=1}^K \Delta u_k a^{x_k} = 0, \\
 \sum_{k=1}^K f_k^3(a, b, c, d) &= \sum_{k=1}^K \Delta u_k (1-x_k) = 0, \\
 \sum_{k=1}^K f_k^4(a, b, c, d) &= \sum_{k=1}^K \Delta u_k x_k = 0
 \end{aligned} \tag{4.49}$$

未定係数 a, b, c, d はこれらの式から一意的に決定することができる .

ところで上記のスプライン曲線の場合には , 4 つの未定係数のうち 2 つは確定で , 残り 2 つも残差の 2 乗の和の微分を 0 と置いた式から問題なく決定された . それは 残差の 2 乗の和を各係数で 2 階微分した式がすべて正になり , このようにして決定した係数が残差の 2 乗の和を最小にするものであったからである . 一方 最小 2 乗指数曲線の場合には , 式 (4.48) を各係数で 2 階微分した式は

$$\frac{2}{a^2} \sum_{k=1}^K [(a^{x_k} b x_k)^2 - \Delta u_k a^{x_k} b (1-x_k) x_k], \quad 2 \sum_{k=1}^K a^{2x_k}, \quad 2 \sum_{k=1}^K (1-x_k)^2, \quad 2 \sum_{k=1}^K x_k^2$$

のようになり , 第 2 式以降は正であるが第 1 式は必ずしも正でない . それは $a^{x_k}, 1-x_k, x_k > 0, 1-x_k$ と x_k の平均的大きさは Δu_k に比べて十分に大きく , 平均的に $|ba^{x_k}| > |\Delta x_k|$ ならば第 1 式も正になると言う意味においてである .

式 (4.49) は係数 a, b, c, d に関し非線形で , a, b, c, d は式 (4.49) から直接求められないので , 予測子修正子法で求めることにする . 予測値は解にある程度近いものを与えるのが無難である . ここではまず与えられたデータを参考に代表的 4 点 $(x, u) = (0, U_1), (1/3, U_2), (2/3, U_3), (1, U_4)$ を取り , これらの点を通る式 (4.47) の未定係数を求める . 簡単のため式 (4.47) と等価な式

$$u(x) = U_1 + (A^{3x} - 1)B + 3xC \tag{4.50}$$

を考える . その 1 階差分は $U_2 - U_1 = (A-1)B + C, U_3 - U_2 = (A-1)AB + C, U_4 - U_3 = (A-1)A^2B + C$, また 2 階差分は $U_3 - 2U_2 + U_1 = (A^2 - 2A + 1)B, U_4 - 2U_3 + U_2 = (A^2 - 2A + 1)AB$ となる . これより式 (??) の係数は $A = (U_4 - 2U_3 + U_2) / (U_3 - 2U_2 + U_1), B = (U_3 - 2U_2 + U_1) / (A^2 - 2A + 1), C = U_2 - U_1 - (A-1)B$ のように求められる . 式 (4.47) と式 (4.50) の比較により , 式 (4.47) の係数の予測値は , $a = A^3, b = B, c = U_1 - B, d = c + 3C$ のように決定できる .

修正値は係数 a, b に関しては式 (4.49) の第 1 , 2 式を Newton 法で解いて , また c, d に関してはその第 3 , 4 式を連立 1 次方程式として解いて求めることができる . Newton 法の式は , 式 (4.49) の第 1 , 2 式に $a = \bar{a} + \Delta a, b = \bar{b} + \Delta b$ を代入し 1 次の大きさの項まで取ったもので , 次のようになる .

$$C_1^i \Delta a + C_2^i \Delta b = RHS^i \quad (i = 1, 2) \tag{4.51a}$$

ただし

$$\begin{aligned}
 C_1^1 &= \sum_{k=1}^K \left\{ \bar{a}^{2x_k-1} x_k^2 \bar{b} + \frac{x_k}{\bar{a}} \bar{f}_k^1 \right\}, & C_2^1 &= \sum_{k=1}^K \bar{a}^{2x_k} x_k, \\
 C_1^2 &= \sum_{k=1}^K \left\{ \bar{a}^{2x_k-1} x_k \bar{b} + \frac{x_k}{\bar{a}} \bar{f}_k^2 \right\}, & C_2^2 &= \sum_{k=1}^K \bar{a}^{2x_k}, \\
 RHS^i &= - \sum_{k=1}^K \bar{f}_k^i \quad (i=1, 2), & \bar{f}_k^i &= f_k^i(\bar{a}, \bar{b}, c, d)
 \end{aligned} \tag{4.51b}$$

また c, d の連立 1 次方程式は次のようになる .

$$C_3^i c + C_4^i d = RHS^i \quad (i=3, 4) \tag{4.52a}$$

ただし

$$\begin{aligned}
 C_3^3 &= \sum_{k=1}^K (1-x_k)^2, & C_4^3 &= \sum_{k=1}^K x_k(1-x_k), & RHS^3 &= \sum_{k=1}^K (u_k - a^{x_k} b)(1-x_k), \\
 C_3^4 &= \sum_{k=1}^K (1-x_k)x_k, & C_4^4 &= \sum_{k=1}^K x_k^2, & RHS^4 &= \sum_{k=1}^K (u_k - a^{x_k} b)x_k
 \end{aligned} \tag{4.52b}$$

次に最小 2 乗指数曲線を求めるプログラムを示す . このプログラムでは前記の最小 2 乗スプラインと同じデータを用いるが全 1 区間とする .

```

***** Interpolation by least square exponential curve
SUBROUTINE LSEXP(x,u,kf,un,nf,u1,u2,u3,u4)
DIMENSION x(kf),u(kf),un(0:nf)
DATA naf,e/10000,1.E-5/
! Predict a...d from 4 typical point data
cA=(u2-2.*u3+u4)/(u1-2.*u2+u3)           !compute A...D
cB=(u1-2.*u2+u3)/(cA-1.)/(cA-1.)
cC=u2-u1-cB*(cA-1.)
a=cA*cA*cA; b=cB; c=u1-cB; d=c+3.*cC     !compute a...d
WRITE(20,'(I10,E15.3,4F10.5)')na,amax,a,b,c,d !print predictor a...d
!a=.13533; b=.5; c=.5; d=.5
na=0; 10 na=na+1
! Correct a and b
C11=0.; C12=0.; C21=0.; C22=0.; RHS1=0.; RHS2=0.
DO k=1,kf; x0=x(k); x1=1.-x0
  ax=a**x0; xx=ax*x0
  du=ax*b+c*x1+d*x0-u(k)                 !compt Delta u
  f1=du*xx; f2=du*ax
  C11=C11+xx*xx*b/a+x0/a*f1; C12=C12+ax*xx; RHS1=RHS1-f1 !setup coefs & rhs
  C21=C21+xx*ax*b/a+x0/a*f2; C22=C22+ax*ax; RHS2=RHS2-f2
ENDDO
del=C11*C22-C21*C12
da=(RHS1*C22-RHS2*C12)/del                !Delta a
db=(C11*RHS2-C21*RHS1)/del                !Delta b
al=1.; IF(na>10)al=1.5
a=a+al*da; b=b+al*db                       !correct a and b
! Correct c and d
C33=0.; C34=0.; C43=0.; C44=0.; RHS3=0.; RHS4=0.

```

```

DO k=1,kf; x0=x(k); x1=1.-x0
  abu=u(k)-a**x0*b
  C33=C33+x1*x1; C34=C34+x1*x0; RHS3=RHS3+abu*x1      !setup coefs & rhs
  C43=C43+x0*x1; C44=C44+x0*x0; RHS4=RHS4+abu*x0
ENDDO
del=C33*C44-C43*C34
dc=(RHS3*C44-RHS4*C34)/del-c      !Delta c
dd=(C33*RHS4-C43*RHS3)/del-d      !Delta d
al=1.; IF(n>10)al=1.5
c=c+al*dc; d=d+al*dd      !correct c and d
! Decide convergence
aa=ABS(da/a); ab=ABS(db/b); ac=ABS(dc/c); ad=ABS(dd/d)
AMAX=AMAX1(aa,ab,ac,ad)
WRITE(20,'(I10,E15.3,4F10.5)')na,amax,a,b,c,d      !print corrector a...d
IF(na<naf.AND.amax>e)      GOTO 10      !decide convergence
! Compute ls-exponent line
DO n=0,nf
  xi=FLOAT(n)/FLOAT(nf); un(n)=a**xi*b+c*(1.-xi)+d**xi      !compute ls-exponential line
ENDDO
END

```

サブルーチン LSEXP のみ示したが、メインプログラムでは予測値のための4点のデータ $U_1 = 1.0$, $U_2 = 0.76$, $U_3 = 0.64$, $U_4 = 0.57$ を与えている。LSEXP ではまず式 (4.50) の下の文章内の式から $ABCD$ それから $abcd$ の予測値を計算し、次いで式 (4.51) から a, b の修正値、式 (4.52) から c, d の修正値を計算し、これらの修正値が十分収束するまで計算を反復し、最後に式 (4.47) から最小2乗指数曲線を計算している。

下表はもとの指数曲線、予測値の指数曲線、最小2乗指数曲線の係数 a, b, c, d の値を示したものである。意外なことにこれらの曲線の係数はかなり違うものになり、解は緩慢に収束し収束解は1995回の反復で得られた。なお予測値としてもとの指数曲線の係数を与えれば収束解は5135回の反復で得られる。下図は最小2乗スプライン曲線と同じグラフィックプログラムを用いて描かれたものである。

	a	b	c	d
$.5 + .5 \exp(-2x)$.13533	.50000	.50000	.50000
予測値	.07234	.35265	.64735	.54449
最小2乗指数曲線	.05130	.05130	.33099	.54672




図 4.4: 与えられたデータと最小2乗指数曲線